# Maintainer: Robin Candau <antiz@archlinux.org>

pkgname='python-omemo-dr'
_pkgname="${pkgname#python-}"
pkgver=1.0.1
pkgrel=1
pkgdesc="OMEMO Crypto Library"
arch=('loong64' 'x86_64')
url="https://dev.gajim.org/gajim/omemo-dr"
license=('GPL3')
depends=('python-cryptography' 'python-protobuf')
makedepends=('python-setuptools' 'python-build' 'python-installer' 'python-wheel')
source=("${url}/-/archive/v${pkgver}/${_pkgname}-v${pkgver}.tar.gz")
sha256sums=('3e7291d3d8b82059d1e26e0c6c4e18c5ca262da84e8ff19ebdf73c2cf3304338')

build() {
	cd "${_pkgname}-v${pkgver}"
	python -m build --wheel --no-isolation
}

package() {
	cd "${_pkgname}-v${pkgver}"
	python -m installer --destdir="${pkgdir}" dist/*.whl
}
